
METAFONT �

Odpowied¹ na konkurs ogªoszony

w biuletynie GUST-u nr 2, str. 27

Bogusªaw Licho«ski

Przedstawienie ramki

Oto zgªoszona przeze mnie ramka:

3A666666666A4

8 9

1 7

1 7

1 7

8 9

0B222222222B5

Do wykre±lenia powy»szej ramki zaprojekto-

waªem dwana±cie nast¦puj¡cych znaków:

0 3 4 5

8 A 9 B

1 6 7 2

Program METAFONT-owy rozpoczyna si¦

(poza mode_setup) od nadania warto±ci parame-

trom:

rw#, rh# okre±laj¡ szeroko±¢ pudeªka w jakim

umieszcz¦ znaki fontu (w przykªadzie pudeªko

ma wymiary 10 � 10 pt)

q parametr okre±laj¡cy odlegªo±¢ dwóch piono-

wych lub poziomych linii ramki (w przykªadzie

q = 2 pt)

rw#=10pt#; % szeroko±¢ pudeªka

rh#=10pt#; % wysoko±¢ pudeªka

q=2pt; % odlegªo±¢ mi¦dzy liniami

De�nicj¦ pierwszego znaku rozpoczynam od:

beginchar("3",rw#,rh#,0);

b¦dzie to lewy górny naro»nik ramki.

Deklaruj¦ cztery u»yteczne ±cie»ki, które

posªu»¡ mi do wyznaczania punktów kontrolnych.

path a; path b; path ml; path mll;

Kolejny krok to podanie wspóªrz¦dnych (x; y)

punktów z1, z2, z3, z6, z11

x1=x11=0.5w;

x3=x1+q;

x2=x6=2x1;

y2=y11=0.5h;

y6=y2-q;

y1=y3=0h;

oraz de�nicja czterech u»ytecznych ±cie»ek.

ml=(z1{right}..{up}z2..(0.5w,h));

mll=(z1{left}..(0,0.5h)..{down}z2);

a=((0.5w,0.5h)+(q,0)--z3);

b=(z6--(0,0.5h)-(0,q));

Przypatrzmy si¦ teraz gotowemu ju» znakowi

3 (lewy górny naro»nik ramki):

Zauwa»my, »e brakuje nam punktów z4, z5, z7,

z8, z9, z10.

Pierwsze dwa nieznane jeszcze punkty wyzna-

cz¦ jako przeci¦cie (dla punktu z4) ±cie»ki a ze

±cie»k¡ ml dla q � 0 lub a ze ±cie»k¡ mll dla q < 0;

if q>=0: z4=a intersectionpoint ml;

else: z4=a intersectionpoint mll;

fi;

punkt z5 wyznaczam analogicznie

if q>=0: z5=b intersectionpoint ml;

else: z5=b intersectionpoint mll;

fi;



Aby wyznaczy¢ punkt z7 de�niuj¦ dwa punkty

pomocnicze (prawy-dolny i lewy-górny punkt

pudeªka) z8 i z9.

z8=(w,0);

z9=(0,h);

Punkt z7 (±rodek na maªym ªuku) wyznaczam jako

przeci¦cie ±cie»ki z8--z9 ze ±cie»k¡ ml.

z7=(z8--z9) intersectionpoint ml;

Kolejny punkt z10 (±rodek na du»ym ªuku)

wyznaczam jako przeci¦cie ±cie»ki z8--z9 ze

±cie»k¡ z1{left}..{down}z2

z10=(z8--z9) intersectionpoint

(z1{left}..{down}z2);

co ko«czy de�nicje punktów kontrolnych. Pozostaªo

nam jedynie zde�niowa¢ odpowiednie piórko:

pickup pencircle scaled 0.01pt;

i zaczynamy rysowanie:

draw z3--z4..z7..z5--z6;

draw z1{left}..z10..{down}z2;

Dorysujmy jeszcze krop¦ w ±rodku, ale

grubszym piórkiem.

pickup pencircle scaled 4pt;

drawdot z11;

Na koniec zapisuj¦ aktualnie utworzony

obrazek pod zmienn¡ LewyGorny i ko«cz¦ de�nicj¦

znaku.

picture LewyGorny;

LewyGorny=currentpicture;

endchar;

Trzy pozostaªe znaki naro»nikowe 0 4 5

de�niuje si¦ analogicznie. S¡ to symetrie osiowe

powy»szego znaku (patrz listing).

Kolejne cztery znaki 8 A 9 B s¡ ju» du»o

ªatwiejsze do zde�niowania. We¹my jeden z nich:

de�niuj¦ wi¦c punkty

z1=(0.5w,0);

if q>0: z2=(0.2w,0.5h)

else: z2=(0.2w+0.6w,0.5h)

fi;

z3=(0.5w,h);

z4=(0.5w+q,0);

z5=(0.5w+q,h);

i prosz¦ METAFONT o narysowanie:

pickup pencircle scaled 0.1pt;

draw z1..z2..z3;

draw z4--z5;

Ko«cówka nie wymaga ju» wyja±nie«:

picture LewyPionowyBrzuszek;

LewyPionowyBrzuszek=currentpicture;

endchar;

wyja±nienia wymaga jednak komenda warunkowa:

if q>0: z2=(0.2w,0.5h)

else: z2=(0.2w+0.6w,0.5h)

fi;

stosuj¦ tu tak¡ instrukcj¦ w celu ÿodwrócenia

brzuszka" w przypadku, gdy q jest ujemne (patrz

poni»szy przykªad).

Trzy pozostaªe znaki ÿgrzuszki" 0 4 5

de�niuje si¦ analogicznie. S¡ to symetrie osiowe

powy»szego znaku lub obrót (patrz listing).

Pozostaªy jeszcze cztery najªatwiejsze do

zde�niowania znaki: 1 6 7 2. Oto jeden z nich:

i jego peªna de�nicja:

bot z1=(0.5w,0);

bot z2=(0.5w+q,0);

top z3=(0.5w,h);

top z4=(0.5w+q,h);

pickup pencircle scaled 0.1pt;

draw z1--z3;

draw z2--z4;

labels(1,2,3,4);

picture LewyPionowySlupek;

LewyPionowySlupek=currentpicture;

Przykªady

Z powy»szych znaków mo»na ÿzªo»y¢" ró»ne ramki.

Pierwsza grupa ramek dla warto±ci:

q=2pt;

mo»e wygl¡da¢ tak:

36666666664

1 7

1 7

02222222225

3A666666666A4

8 9

1 7

1 7

8 9

0B222222222B5

Wystarczy jednak zmieni¢ parametr q tak oto:

q=5pt;

aby otrzyma¢:

36666666664

1 7

1 7

02222222225

3A666666666A4

8 9

1 7

1 7

8 9

0B222222222B5

Zmieniaj¡c po raz kolejny parametr q na (q

mniejsze od zera):

q=-2pt;

otrzymujemy

y

:

36666666664

1 7

1 7

02222222225

3A666666666A4

8 9

1 7

1 7

8 9

0B222222222B5

Mam nadziej¦, i» moja ramka spodoba si¦

Wysokiemu Jury.

3A666666666A4

8 9

1 7

1 THE END 7

1 7

8 9

0B222222222B5

Listing

%%%%% Plik ramka.mf wersja 2.6 %%%%%

mode_setup;

rw#=10pt#;

rh#=10pt#;

q=2pt;

beginchar("3",rw#,rh#,0);

path a; path b; path ml; path mll;

x1=x11=0.5w;

x3=x1+q;

x2=x6=2x1;

y2=y11=0.5h;

y6=y2-q;

y1=y3=0h;

ml=(z1{right}..{up}z2..(0.5w,h));

mll=(z1{left}..(0,0.5h)..{down}z2);

a=((0.5w,0.5h)+(q,0)--z3);

b=(z6--(0,0.5h)-(0,q));

if q>=0: z4=a intersectionpoint ml;

else: z4=a intersectionpoint mll;

fi;

if q>=0: z5=b intersectionpoint ml;

else: z5=b intersectionpoint mll;

fi;

z8=(w,0);

y: Zwró¢my uwag¦ na odwrócone brzuszki.



z9=(0,h);

z7=(z8--z9) intersectionpoint ml;

z10=(z8--z9) intersectionpoint

(z1{left}..{down}z2);

pickup pencircle scaled 0.01pt;

draw z3--z4..z7..z5--z6;

draw z1{left}..z10..{down}z2;

pickup pencircle scaled 4pt;

drawdot z11;

labels(1,2,3,4,5,6,7,8,9,10,11);

picture LewyGorny;

LewyGorny=currentpicture;

endchar;

%---------------------------------

beginchar("4",rw#,rh#,0);

currentpicture:=LewyGorny

reflectedabout((0.5w,0),(0.5w,h));

endchar;

%---------------------------------

beginchar("0",rw#,rh#,0);

currentpicture:=LewyGorny

reflectedabout((0,0.5h),(w,0.5h));

picture PrawyDolny;

PrawyDolny=currentpicture;

endchar;

%---------------------------------

beginchar("5",rw#,rh#,0);

currentpicture:=PrawyDolny

reflectedabout((0.5w,0),(0.5w,h));

endchar;

%=================================

beginchar("8",rw#,0.6rh#,0);

z1=(0.5w,0);

if q>0: z2=(0.2w,0.5h)

else: z2=(0.2w+0.6w,0.5h)

fi;

z3=(0.5w,h);

z4=(0.5w+q,0);

z5=(0.5w+q,h);

pickup pencircle scaled 0.1pt;

draw z1..z2..z3;

draw z4--z5;

labels(1,2,3,4,5);

picture LewyPionowyBrzuszek;

LewyPionowyBrzuszek=currentpicture;

endchar;

%---------------------------------

beginchar("9",rw#,0.6rh#,0);

currentpicture:=LewyPionowyBrzuszek

reflectedabout((0.5w,0),(0.5w,0.6h));

endchar;

%---------------------------------

beginchar("B",0.6rw#,rh#,0);

currentpicture:=LewyPionowyBrzuszek

rotatedabout((0.5w,0.3h),90);

picture DolnyPoziomyBrzuszek;

DolnyPoziomyBrzuszek=currentpicture;

endchar;

%---------------------------------

beginchar("A",0.6rw#,rh#,0);

currentpicture:=DolnyPoziomyBrzuszek

reflectedabout((0,0.5h),(w,0.5h));

endchar;

%=================================

beginchar("1",rw#,rh#,0);

bot z1=(0.5w,0);

bot z2=(0.5w+q,0);

top z3=(0.5w,h);

top z4=(0.5w+q,h);

pickup pencircle scaled 0.1pt;

draw z1--z3;

draw z2--z4;

labels(1,2,3,4);

picture LewyPionowySlupek;

LewyPionowySlupek=currentpicture;

endchar;

%---------------------------------

beginchar("7",rw#,rh#,0);

currentpicture:=LewyPionowySlupek

reflectedabout((0.5w,0),(0.5w,h));

picture PrawyPionowySlupek;

PrawyPionowySlupek=currentpicture;

endchar;

%---------------------------------

beginchar("6",rw#,rh#,0);

currentpicture:=PrawyPionowySlupek

rotatedabout((0.5w,0.5h),90);

picture GornyPoziomySlupek;

GornyPoziomySlupek=currentpicture;

endchar;

%---------------------------------

beginchar("2",rw#,rh#,0);

currentpicture:=GornyPoziomySlupek

reflectedabout((0,0.5h),(w,0.5h));

endchar;

end.

Ps. Od Redakcji

Zaprezentowany wy»ej artykuª zwyci¦»yª w naszym

konkursie. Nagrod¡ jest ksi¡»ka N. Walsha:

Making T

E

X Work, któr¡ przekazujemy Bogusiowi

Licho«skiemu. Gratulujemy! Listing programu

METAFONT-owego mo»na znale¹¢ w archiwum

sunrise.pg.gda.pl: /TeX/GUST/bulletin/05.


